

Raport z Projektu WSYZ

Wojciech Kołodziejak

Mateusz Maj

Aleksandra Majewska

Bartłomiej Rasztabiga

12 czerwca 2022

1 Treść zadania

Rozważana jest produkcja i dystrybucja podstawowych warzyw, tj. ziemniaków, kapusty, buraków i marchwi w Warszawie i okolicach. Istnieją trzy rodzaje przedsiębiorstw:

- Grupa 6 producentów: P1...P6. Każdy z producentów produkuje każdy rodzaj warzyw jednak w różnych maksymalnych ilościach rocznych. Lokalizacja producentów to: Błonie, Książenice, Góra Kalwaria, Otwock, Wołomin, Legionowo.
- Sieć 3 magazynów-chłodni: M1..M3. Każdy magazyn ma określoną pojemność wyrażoną w tonach (800, 1200, 750) i może służyć do przechowywania dowolnych warzyw. Lokalizacje magazynów to Pruszków, Piaseczno, Zielonka.
- Sieć sklepów spożywczych usytuowanych w Warszawie.

Tablica 1: Producenci

nazwa	adres
P1	Błonie
P2	Książenice
P3	Góra Kalwaria
P4	Otwock
P5	Wołomin
P6	Legionowo

Tablica 2: Magazyny


nazwa	adres
M1	Pruszków
M2	Piaseczno
M3	Zielonka

Tablica 3: Sklepy

nazwa	adres	współrzędne	pojemność magazynu
S1	Béli Bartóka 8, 02-787 Warszawa	(52.162260, 21.028860)	7,10 t
S2	Sady Żoliborskie 2, 01-772 Warszawa	(52.267020, 20.974340)	7,18 t
S3	al. Jerozolimskie 184, 02-486 Warszawa	(52.200650, 20.932470)	7,19 t
S4	Sportowa 30, 05-090 Raszyn	(52.150160, 20.930130)	7,22 t
S5	Jana Pawła Woronicza 17, 00-999 Warszawa	(52.188710, 21.011170)	7,18 t

S6	Aleksandra Gierymskiego 19, 00-772 Warszawa	(52.200830, 21.034420)	7,23 t
S7	Jana Nowaka-Jeziorańskiego 44/U1, 03-982 Warszawa	(52.230050, 21.078680)	7,29 t
S8	plac Stanisława Małachowskiego 2, 00-066 Warszawa	(52.238890, 21.012850)	7,18 t
S9	Jana Ciszewskiego 15, 02-777 Warszawa	(52.153782, 21.039921)	7,21 t
S10	Górczewska 88, 01-117 Warszawa	(52.239310, 20.945250)	7,23 t

Na mapie oznaczono kolejno producentów, magazyny i sklepy




2 Model biznesowy


Model biznesowy w notacji BPMN został zaprojektowany w aplikacji Bizagi Modeler.

// TODO opis modelu i kolejnych sekcji


Rysunek 1: BPMN 1




Rysunek 2: BPMN 2




Rysunek 3: BPMN 3



Rysunek 4: BPMN 4



Rysunek 5: BPMN 5



3 Model optymalizacyjny

3.1 Opis modelu

Nasz model składa się z:

- 4 zbiorów
- 8 parametrów
- 3 zmiennych decyzyjnych
- 7 ograniczeń
- funkcji celu

Zbiory:

- producenci (P1 - P6)
- magazyny (M1 - M3)
- warzywa (ziemniaki, kapusta, buraki, marchew)
- sklepy (S1 - S10)

Parametry:

- tygodnie w roku (52)
- podaż producentów
- maksymalna pojemność magazynów
- odległości od producentów do magazynów
- odległości od magazynów do sklepów
- prognozowana sprzedaż warzyw (dla każdego tygodnia)
- maksymalna pojemność przysklepowych magazynów
- koszt transportu 1 tony na odległość 1 kilometra

Zmienne decyzyjne:

- coroczny transport od producentów do magazynów (ile producenci muszą wyprodukować)
- cotygodniowy transport od magazynów do sklepów
- cptygodniowy stan zapasu warzyw w magazynach przysklepowych

Ograniczenia:

- stan zapasu warzyw w sklepie jest z tygodnia na tydzień zmniejszany o prognozę sprzedaży i zwiększany o cotygodniowy transport do sklepu
- zapas warzyw w sklepie nie może przekroczyć maksymalnej pojemności przysklepowego magazynu
- cotygodniowy transport warzyw nie może przekroczyć maksymalnej pojemności przysklepowego magazynu
- minimalny zapas warzyw sklepie musi minimalnie wynosić 10% prognozy sprzedaży
- coroczny transport do magazynów musi być większy lub równy od transportu do sklepów
- coroczny transport do magazynów musi być mniejszy lub równy podaży producentów
- coroczny transport do magazynów musi być mniejszy lub równy maksymalnej pojemności magazynów

3.2 Zapis modelu w AMPL

```
set PRODUCENTS;
set WAREHOUSES;
set VEGETABLES;
set STORES;

param T > 0;
param supply {PRODUCENTS,VEGETABLES} > 0;
param max_warehouse_capacity {WAREHOUSES} >= 0;
param distance_to_warehouse {PRODUCENTS,WAREHOUSES} > 0;
param distance_to_store {WAREHOUSES,STORES} > 0;
param weekly_sales_forecast {1..T, STORES, VEGETABLES} >= 0;
param store_warehouse_capacity {STORES} >= 0;
param km_cost > 0;

var yearly_transport_to_warehouses {PRODUCENTS,WAREHOUSES,VEGETABLES} >= 0;
var weekly_transport_to_stores {1..T,WAREHOUSES,STORES,VEGETABLES} >= 0;
var weekly_stores_warehouse_stock {1..T,STORES,VEGETABLES} >= 0;

minimize Total_Cost:
    sum {p in PRODUCENTS, w in WAREHOUSES, v in VEGETABLES}
        distance_to_warehouse[p,w] * km_cost *
        yearly_transport_to_warehouses[p,w,v]
    +
    sum {w in WAREHOUSES, s in STORES, v in VEGETABLES, n in 1..T}
        distance_to_store[w,s] * km_cost * weekly_transport_to_stores[n,w,s,v];

subject to Store_Warehouse_Demand {s in STORES, n in 2..T, v in VEGETABLES}:
    weekly_stores_warehouse_stock[n, s, v] = weekly_stores_warehouse_stock[n-1,
        s, v] - weekly_sales_forecast[n, s, v] + sum {w in WAREHOUSES}
        weekly_transport_to_stores[n, w, s, v];

subject to Store_Warehouse_Demand_1_week {s in STORES, v in VEGETABLES}:
    weekly_stores_warehouse_stock[1, s, v] = - weekly_sales_forecast[1, s, v] +
        sum {w in WAREHOUSES} weekly_transport_to_stores[1, w, s, v];

subject to Store_Warehouse_Max_Capacity {s in STORES, n in 1..T}:
    sum {v in VEGETABLES} weekly_stores_warehouse_stock[n, s, v] <=
        store_warehouse_capacity[s];

subject to Store_Warehouse_Min_Capacity {s in STORES, n in 1..T, v in
    VEGETABLES}:
```

```

weekly_stores_warehouse_stock[n, s, v] >= 0.1 * weekly_sales_forecast[n, s,
v];

subject to Store_Warehouse_Max_Capacity_Transport {s in STORES, n in 1..T}:
    sum {w in WAREHOUSES, v in VEGETABLES} weekly_transport_to_stores[n, w, s, v]
    <= store_warehouse_capacity[s];

subject to Warehouse_Supply {w in WAREHOUSES, v in VEGETABLES}:
    sum {p in PRODUCENTS} yearly_transport_to_warehouses[p, w, v] >= sum {s in
STORES, n in 1..T} weekly_transport_to_stores[n, w, s, v];

subject to Producent_Supply {p in PRODUCENTS, v in VEGETABLES}:
    sum {w in WAREHOUSES} yearly_transport_to_warehouses[p,w,v] <= supply[p, v];

subject to Warehouse_Max_Capacity {w in WAREHOUSES}:
    sum {p in PRODUCENTS, v in VEGETABLES} yearly_transport_to_warehouses[p,w,v]
    <= max_warehouse_capacity[w];


```

4 Wyniki i wnioski

Dla ustalonych danych wejściowych (transp.dat) całkowity minimalny koszt transportu wyniósł: 248 837 zł.

Minimalna suma pojemności magazynów do rozwiązania problemu wynosi 1881 ton. Poniżej tej wartości nie da się znaleźć rozwiązania dla zadanego popytu (prognozy sprzedaży). Oryginalnie (z treści zadania) wynosiła ona 2750 ton.

Poniżej wyniki symulacji (koszt transportu), jeżeli cały towar byłby transportowany do/z tylko jednego magazynu.



Jak widać, rozkład transportu na różne magazyny ma sens. Koszt jest wtedy minimalny, ponieważ sklepy pobierają towar z magazynu znajdującego się najbliżej.

Analizując wynik optymalizacji (konkretnie zmienną „weekly transport to stores”) można zauważyć, że niektóre magazyny w ogóle nie zaopatrują niektórych sklepów (np. M1 nie zaopatruje w buraku S5, S6, S7 i S9). Jest to spowodowane ich dużą wzajemną odległością, czyli dużym kosztem transportu.

```

weekly_transport_to_stores [*,M1,*,Buraki]
:   S1    S10    S2    S3    S4    S5    S6    S7    S8    S9    :=
1   0    5.195    2.194    1.254    1.463    0    0    0    1.364    0
2   0    4.341    0.281    1.085    1.11    0    0    0    1.152    0
3   0    1.142    1.816    2.501    1.196    0    0    0    2.319    0
4   0    1.261    3.04    0    2.419    0    0    0    0.146    0
5   0    0.152    0.117    1.113    0.95    0    0    0    1.254    0
6   0    0    0    2.398    0    0    0    0    2.725    0
7   0    0    1.247    2.359    4.799    0    0    0    0.738    0
8   0    0    1.184    0    1.268    0    0    0    0    0
9   0    0    4.592    0    0    0    0    0    1.323    0
10  0    0    0    1.123    0    0    0    0    1.134    0
11  0    1.144    0    2.02    0    0    0    0    4.971    0
12  0    0.16    0    0    0.304    0    0    0    0    0
13  0    2.419    1.966    0    0.28    0    0    0    0.445    0
14  0    1.481    2.124    0    0.279    0    0    0    0    0
15  0    0    0.25    0.274    2.895    0    0    0    0    0
16  0    0    0    0.801    0    0    0    0    0    0
17  0    2.99    0    1.33    0    0    0    0    0    0
18  0    0.788    0    0    0    0    0    0    0    0
19  0    0.7    0    0    0    0    0    0    0    0
20  0    0.655    0    0    0    0    0    0    0    0
21  0    0    0    0    0    0    0    0    0    0
22  0    0    0.926    0    0    0    0    0    0    0
23  0    0    0.867    0    0    0    0    0    0    0
24  0    0    0    0.619    0.275    0    0    0    0    0
25  0    0    0    0    0.236    0    0    0    0    0
26  0    0    2.449    1.249    6.942    0    0    0    4.959    0
27  0    0.189    0    1.292    0.815    0    0    0    0    0
28  0    0    2.37    2.432    1.255    0    0    0    0    0
29  0    0    0    0    0    0    0    0    0    0
30  0    0.985    0.946    1.124    0    0    0    0    1.158    0
31  0    2.482    0    1.085    0    0    0    0    1.632    0
32  0    0    0    1.112    1.556    0    0    0    2.478    0
33  0    1.207    1.158    0.89    0    0    0    0    2.5    0
34  0    1.045    1.148    1.13    0    0    0    0    2.331    0
35  0    1.192    1.084    1.187    0.935    0    0    0    0    0
36  0    1.081    3.709    1.081    3.341    0    0    0    0    0
37  0    1.299    0    1.233    0    0    0    0    0.545    0
38  0    1.137    0    1.132    0    0    0    0    0    0
39  0    0.215    0.224    0.326    0.124    0    0    0    0.074    0
40  0    0.344    0.243    0.29    0.22    0    0    0    1.239    0
41  0    0.23    1.805    0.514    1.286    0    0    0    0    0
42  0    0.339    0    0    1.51    0    0    0    0    0
43  0    0.596    0    0.616    1.152    0    0    0    0    0
44  0    0    0    0    0.515    0    0    0    0.381    0
45  0    0.367    0    0.245    0.8    0    0    0    0.281    0
46  0    2.184    3.73    1.426    0    0    0    0    1.588    0
47  0    1.761    0.585    3.05    0    0    0    0    1.319    0
48  0    0    0.167    0.212    0    0    0    0    0    0
49  0    0    0    0    0    0    0    0    0.853    0
50  0    0    0    0    0    0    0    0    0    0
51  0    0    0    0    0    0    0    0    0    0
52  0    1.156    0    0    1.25    0    0    0    1.201    0

```

// TODO analiza wyniku, co by sie zmienilo, wplyw parametrow na wynik etc.
// TODO wrzucic jakies fajne wykresy (np. zaleznosc kosztu od jakiegos parametru)

5 Harmonogram prac

// TODO dodac osoby?

- Zapoznanie się z problemem (31.03-3.04)
- Utworzenie modelu BPMN (3.04-6.04)
- Recenzja i poprawki do modelu BPMN (6.04-8.04)
- Ustalenie danych wejściowych do modelu optymalizacyjnego: lokalizacje sklepów (15.04-22.04)
- Uszczegółowienie danych wejściowych do modelu optymalizacyjnego: prognoza sprzedaży, maksymalne pojemności magazynów itd. (22.04-29.04)
- Utworzenie modelu optymalizacyjnego w języku AMPL (29.04-6.05)
- Testy rozwiązania przed oddaniem etapu (6.05-13.05)
- Poprawki do modelu optymalizacyjnego, dodano między innymi śledzenie stanu magazynów przysklepowych (20.05-27.05)
- Finalne testy rozwiązania (27.05-3.06)
- Utworzenie raportu z projektu (3.06-13.06)